Alk Primer

(Draft)

Dorel Lucanu

April 10, 2016

Chapter 1

Introduction

1.1 Motivation

Alk is an algorithmic language intended to be used for teaching data structures and algorithms using an abstraction notation (independent of programming language).

The goal is to have a language that:

- is simple to be easily understood;
- is expressive enough to describe a large class of algorithms from various problem domains;
- is abstract: the algorithm description must make abstraction of implementation details, e.g., the low-level representation of data;
- is a good mean for learning how to algorithmically think;
- supply a rigorous computation model suitable to analyse algorithms;
- is executable: the algorithm can be executed, even if they are partially designed;
- is accompanied by a set of tools helping to analyse the algorithm correctness and the efficiency;
- input and output are given as abstract data types, ignoring implementation details.

As a starting example we consider the Alk description of the Euclid algorithm:

```
gcd(a, b)
{
  while (a != b) {
    if (a > b)  a = a - b;
    if (b > a) b = b - a;
  }
  return a;
}
```

The algorithm is described using a C++-like notation. The name of the alghorithm is `gcd` and its input parameters are `a` and `b`. There is no need to declare the type of parameters and/or the type of the return value. In order to execute the `gcd` algorithm, just add a single line algorithm

```
x = gcd(12, 8);
```

and execute it (gcd.alk is the file include the above code):

```
> alk gcd.alk
    x |-> 4
```

The output is the final configuration of the execution, that includes the values of the global variables (here `x`). An alternative is to write a general call of the algorithm

```
x = gcd(u, v);
```

and execute it by mentioning the initial values of the global variables `u` and `v`:

```
> alk gcd.alk -init="u |-> 42 v |-> 56"
The state from the final configuration:
    u |-> 42
    v |-> 56
    x |-> 14
```

A more complex algorithm is the DFS traversal of a digraph represented with external adjacent lists:

```
dfsRec(i, out S) {
  if (S[i] == 0) {
    // visit i
    S[i] = 1;
    p = (D.a)[i];
    while (p.size() > 0) {
      j = p.topFront();
      p.popFront();
      dfsRec(j, S);
    }
  }
}

// the calling algorithm
dfs(D, i0) {
  i = i0;
  while (i < D.n) {
    S[i] = 0;
    i = i + 1;
  }
  dfsRec(i0, S);
  return S;
}
reached = dfs(D, i0);
```

The execution of the above algorithm on the digraph:

$$
\begin{aligned}
&D.n = 3,\\
&D.a[0] = \langle 1, 2\rangle\\
&D.a[1] = \langle 2, 0\rangle\\
&D.a[2] = \langle 0\rangle
\end{aligned}
$$

is obtained as follows:

```
> alk dfsrec.alk  -cINIT="D |-> { n -> 3
                           a -> [ < 1, 2 >, < 2, 0 >,  < 0 > ] }
                           i0 |-> 1"
The state from the final configuration:
    D |-> { (n -> 3) (a -> ([ (< 1, 2 >), (< 2, 0 >), (< 0 >) ])) }
    i0 |-> 1
    reached |-> [ 1, 1, 1 ]
```

TODO. *Write a Perl prototype for the* `alk` *tool.*

1.2 Executing Alk Algorithms with K Tools

The Alk language was designed using K Framework (`http://www.kframework.org/`). So we can use K tools for executing algorithms. The K definition of Alk is compiled using the `kompile` tool:

```
> kompile alk.k
```

Then the compiled definition can be used to execute Alk algorithms:

```
> krun gcd.alk -cINIT=".Map"
<k>
    .K
</k>
<state>
    x |-> 4
</state>
```

The initial state is given as the value of the option `-cINIT`; `.Map` is the K notation for the empty state. A nonempty state is specified as expected:

```
> krun gcd.alk -cINIT="u |-> 42 v |-> 56"
<k>
    .K
</k>
<state>
    u |-> 42
    v |-> 56
    x |-> 14
</state>
```

Here is the execution of the DFS algorithm with the K Tool:

```
> krun dfsrec.alk  -cINIT="D |-> { n -> 3
                            a -> [ < 1, 2 >, < 2, 0 >,  < 0 > ] }
                            i0 |-> 1"



<k>
    .K
</k>
<state>
    D |-> { (n -> 3) (a -> ([ (< 1, 2 >), (< 2, 0 >), (< 0 >) ])) }
    i0 |-> 1
    reached |-> [ 1, 1, 1 ]
</state>
```

Remark. *This is a in progress document that is incrementally updated.*

Chapter 2

Language Description

2.1 Variables and their Values

Alk includes two categories of values: 3ex

Scalars (primitive values). Here are included the booleans, integers, rationals (floats), and strings.

Structured values. Here are included the sequences (linear lists), arrays, structures.

Note that a data can be as complex as possible, i.e, we may have arrays of sequences, arrays of arrays, sequences of arrays of structures, structures of arrays and lists, and so on.

2.1.1 Scalars

The scalars are written using a syntax similar to that from the most popular programming languages:

```
index = 234;
isEven = true;
radius = 21.468;
name = "john";
```

The execution of the above algorithm produces the following output:

```
index |-> 234
isEven |-> true
radius |-> 2.1468e+01
name |-> "john"
```

Note that for the floating point numbers the scientific notation is used.

2.1.2 Arrays

An array value is written as a sequence surrounded by square brackets: $[v_0, \ldots, v_{n-1}]$, where v_i is a value, for $i = 0, \ldots, n-1$. Here is a very simple algorithm handling arrays:

Algorithm	Output

`a = [3, 5, 6, 4];`
`i = 1;`
`x = a[i];`
`a[i+1] = x;`
	`a |-> [ 3, 5, 5, 4 ]`
	`i |-> 1`
	`x |-> 5`

The multi-dimensional arrays are represented a arrays of arrays:

```
a = [ [ 1, 2, 3 ], [ 4, 5, 6 ] ];
b = a[1];
c = a[1][2];
a[0] = b;
a[1][1] = 89;

w = [ [ [ 1, 2], [ 3, 4 ] ], [ [ 5, 6 ], [ 7, 8 ] ] ];
x = w[1];
y = w[1][0];
z = w[1][0][1];
w[0][1][0] = 99;
```

The output:

```
a |-> [ ([ 4, 5, 6 ]), ([ 4, 89, 6 ]) ]
b |-> ([ 4, 5, 6 ])
c |-> 6
w |-> [ ([ ([ 1, 2 ]), ([ 99, 4 ]) ]), ([ ([ 5, 6 ]), ([ 7, 8 ]) ]) ]
x |-> ([ ([ 5, 6 ]), ([ 7, 8 ]) ])
y |-> ([ 5, 6 ])
z |-> 6
```

2.1.3 Sequences (linear lists)

A sequence value is written in a similar to an array, but using angle brackets: $\langle v_0, \ldots, v_{n-1} \rangle$, where v_i is a value, for $i = 0, \ldots, n-1$. The list of operations over sequences includes:

`empty()`	returns the empty list $\angle\rangle$
L.`topFront()`	returns v_0
L.`topBack()`	returns v_{n-1}
L.`at(`i`)`	returns v_i
L.`insert(`i`,`x`)`	returns $\langle \ldots v_{i-1}, x, v_i, \ldots \rangle$
L.`remove(`i`,`x`)`	returns $\langle \ldots v_{i-1}, v_{i+1}, \ldots \rangle$
L.`size()`	returns n
L.`popFront()`	returns $\langle v_1, \ldots, v_{n-1} \rangle$
L.`popBack()`	returns $\langle v_0, \ldots, v_{n-2} \rangle$
L.`pushFront(`x`)`	returns $\langle x, v_0, \ldots, v_{n-1} \rangle$
L.`pushBack(`x`)`	întoarce $\langle v_0, \ldots, v_{n-1}, x \rangle$
L.`update(`i`,`x`)`	returns $\langle \ldots v_{i-1}, x, v_{i+1}, \ldots \rangle$

Example:

Algorithm	Output

`ll = < 8, 3, 9, 4, 5, 4 >;`
`i = 1;`
`x = ll.at(i + 1);`
`y = ll.topFront();`
`ll.insert(2, 22);`
`ll.update(3, 33);`
`l2 = ll;`
`l2.removeAt(0);`
`l2.removeAt(3);`
`l2.removeAllEqTo(4);`
	`i |-> 1`
	`l2 |-> < 3, 22, 33, 5 >`
	`ll |-> < 8, 3, 22, 33, 4, 5, 4 >`
	`x |-> 9`
	`y |-> 8`

Now we may define sequences of arrays:

Algorithm

```
l = < [1, 2, 3], [4, 5] >;
a = l.at(1);
l.pushBack(a);
```

Output

```
a |-> [ 4, 5 ]
l |-> < ([ 1, 2, 3 ]), [ 4, 5 ], [ 4, 5 ] >
```

and arrays of structures:

Algorithm

```
a = [ < 1, 2, 3 >, < 4, 5, 6 >, < 7, 8 > ];
b = a[1];
a[1].update(1, 11);
a[0].insert(2,22);
a[2] = a[1].pushBack(77);
```

Output

```
a |-> [ (< 1, 2, 22, 3 >), (< 4, 11, 6 >), (< 4, 11, 6, 77 >) ]
b |-> < 4, 5, 6 >
```

2.1.4 Structures

A structure value is of the form $\{f_1 \to v_1 \dots f_n \to v_n\}$, where f_i is a field name and v_i is a value, for $i = 1, \dots, n$.

Example:

Algorithm

```
s = { x -> 12  y -> 45 };
a = s.x;
s.y = 99;
b.x = 22;
```

Output

```
a |-> 12
b |-> { x -> 22 }
s |-> { (x -> 12) (y -> 99) }
```

Note that the structure `b` has been created with only one field, because there is no information about its type, which is deduced on the fly during the execution.

We may have structures of arrays

Algorithm

```
s = { x -> [ 1, 2, 3 ] y -> [ 4, 5, 6 ] };
b = s.y;
(s.x)[1] = 11;
```

Output

```
b |-> [ 4, 5, 6 ]
s |-> { (x -> ([ 1, 11, 3 ])) (y -> [ 4, 5, 6 ]) }
```

sequences of arrays

Algorithm

```
l = < { x -> 12 y -> 56 }, { x -> -43 y -> 98 }, { x -> 33 y -> 66 } >;
u = l.topFront();
l.pushBack({ x -> -100 y -> 200 });
l.at(2).x = 999;
```

Output

```
a |-> [ 4, 5 ]
l |-> < ([ 1, 2, 3 ]), [ 4, 5 ], [ 4, 5 ] >
```

and so on.

TODO. *Change the notation* `x -> v` *into* `x : v`.

2.1.5 Sets

A set value is written as $\{v_0, \ldots, v_{n-1}\}$, where v_i is a value, for $i = 0, \ldots, n-1$. The operations over sets include the union `U`, the intersection `^`, the difference `\`, and the membership test `_ belongsTo _`. Example:

Algorithm	Output

`s1 = { 1 .. 5 };`
`s2 = { 2, 4, 6, 7 };`
`a = s1 U s2 ;`
`b = s1 ^ s2;`
`c = s1 \ s2;`
`t = 2 belongsTo b ^ c;`
	`a |-> { 7, 6, 1, 2, 3, 4, 5 }`
	`b |-> { 2, 4 }`
	`c |-> { 1, 3, 5 }`
	`s1 |-> { 1, 2, 3, 4, 5 }`
	`s2 |-> { 2, 4, 6, 7 }`
	`t |-> false`

Obviously, we may have sets of arrays, sequences of sets, and so on.

Remark. *The current implementation does check if a set value assigned to a variable is indeed a set. But the operations returns sets whenever the arguments are sets.*

2.1.6 Specification of values

Alk includes several sugar syntax mechanisms for specifying values in a more compact way:

Algorithm	Output

`p = 3;`
`q = 9;`
`a = [ i | i : p .. q ];`
`p = 2;`
`b = [ a[i] | i : p .. p+3 ];`
`l = < b[i] * 2 | i : p-2 .. p >;`
	`a |-> [ 3, 4, 5, 6, 7, 8, 9 ]`
	`b |-> [ 5, 6, 7, 8 ]`
	`l |-> < 10, 12, 14 >`
	`p |-> 2`
	`q |-> 9`

2.1.7 Flexibility in Using Data Structures

Even if each data structure has its own operations with specific notations, these operations and notations can be "exported" to the other data structures as sugar syntax. For instance, if `L` is a sequence, then we may use `L`$[i]$ for `L.at`(i).

Example:

Algorithm	Output

`L = < 8, 3, 9, 4 >;`
`x = ll[2];`
`L[3] = 33;`
`L[4] = 44;`
	`L |-> < 8, 3, 9, 33, 44 >`
	`x |-> 9`

If `S` is a set, then we can borrow the notations from sequences and array operations:

`S`$[i]$ – returns an element from S. The same do `S.topFront()` and `S.topBack()`. Note that the behaviour of these operations could be nondeterministic: for the same set vale, they could return different elements in different execution steps;

`S.insert`(i, x) – adds x to S (it is a short notation for `S = S U` (x));

Example:

Algorithm	Output

`s = { 2, 4, 6, 7 };`
`z = s.topFront();`
`n = s.size();`
`s2 = s.pushBack(99);`
`a = s2[3];`
`s.insert(2, 22);`
	`a |-> 7`
	`n |-> 4`
	`s2 |-> { 2, 4, 6, 7, 99 }`
	`s |-> { 2, 4, 22, 6, 7 }`
	`z |-> 2`

Here is an example that shows that `_.topFron()` returns different values, even if it is called for equal sets:

Algorithm	Output

`s = { 2, 4, 6, 7 };`
`s2 = s;`
`a = s.topFront();`
`s2.popFront();`
`s2.pushBack(a);`
`b = s2.topFront();`
`t = s == s2;`
	`a |-> 2`
	`b |-> 4`
	`s2 |-> { 4, 6, 7, 2 }`
	`s |-> { 2, 4, 6, 7 }`
	`t |-> true`

We can see that `s` and `s2` are equal as sets, but `a` and `b` are different.

2.1.8 Lists with iterators

An iterator p associated with a list L if p "refers" an element of L. With iterators one can call operation over the associated lists and/or traverse the associated list.

Operations with ieterators:

p `+` i – returns an iterator refering the ith element after p;

p `-` i – returns an iterator refering the ith element in front of p;

`++`p – moves p to the previous element (if any);

`--`p – moves p to the next element (if any);

L`.first()` – returns an iterator that refers the first element of L;

L`.end()` – returns an invalid iterator for L

Using iterators, one can access the lements of the lists and/or execute operations on the associated list:

`*`p – returns the referred element;

p`->delete()` – remove the element referred by p and move p to the next element;

p`->insert(x)` – insert x immediately after the element referred by p.

If p refers the ith element in L, then p`->delete()` is equivalent to $L[i]$`.delete()` and p`->insert(x)` with $L[i]$`.insert(x)`.

The following operators are useful to traverse circular lists:

p `+%` i – returns an iterator refering the ith element after p modulo the length of the list;

p `-%` i – returns an iterator refering the ith element in front of p modulo the length of the list;

`++%`p – moves p to the previous element modulo the length of the list;

`--%`p – moves p to the next element modulo the length of the list

Example 1:

Algorithm	Output

```
L = < 2, 5, 8 >;
i = 0;
A = [];
for (p = L.first(); p != L.end(); ++p) {
  A[i] = *p ;
  ++i;
}
```

```
A |-> [ 2, 5, 8 ]
L |-> < 2, 5, 8 >
i |-> 3
p |-> it ( L , 3 )
```

Example 2:

Algorithm	Output

```
L = < 2, 5, 8, 33 >;
p = L.first();
p = p + 3;
a = *p;
++% p;
b = *p;
q = L.first();
c = *(--% q);
```

```
L |-> < 2, 5, 8, 33 >
a |-> 33
b |-> 2
c |-> 33
p |-> it ( L , 0 )
q |-> it ( L , 3 )
```

2.2 Expressions

Alk includes the basic operators over scalars with a C++-like syntax.

Since Alk is designed with K Framework, it can be easily extended with new operators.

2.3 Statements

The syntax for the statements is similar to that of imperative C++.

We already have seen examples of the assignment statement. The other statements include:

2.3.1 Block

Syntax: `{` *Stmt* `}`

2.3.2 `if`

Syntax: 1) `if (` *Exp* `)` *Stmt* `else` *Stmt* 2) `if (` *Exp* `)` *Stmt*

2.3.3 `while`

Syntax: `while (` *Exp* `)` *Stmt*

2.3.4 `for`

Syntax: 1) `forall` *Id* `in` *Exp Stmt* 2) `for (` *VarAssign* `;` *Exp* `;` *VarUpdate* `)` *Stmt*

Examples:

```
for (i= 2; i <= x / 2; ++i)
  if (x % i == 0) return false;

forall y in { 1 .. 6 }
  if (y belongsTo s2) d = d U singletonSet(y);
```

2.3.5 Sequential Composition

Syntax: *Stmt Stmt*

2.4 Statements for Nondeterministic Algorithms

2.4.1 `choose`

Syntax: `choose` *Id* `in` *Exp* `;` 2) `choose` Id `in` *Exp* `s.t.` *Exp* `;`

Example 1:

Algorithm	Output
`choose x1 in { 1 .. 5 };` `choose x2 in { 1 .. 5 };`	`x1 |-> 3` `x2 |-> 1`

Example 2:

Algorithm	Output

`odd(x) {`
`  return x % 2 == 1;`
`}`
 	
`choose x1 in { 1 .. 5 } s.t. odd(x1);`
`choose x2 in { 1 .. 5 } s.t. odd(x2);`
	`x1 |-> 5`
	`x2 |-> 3`

2.4.2 `success`

Syntax: `success ;`

Example:

Algorithm	Output

`odd(x) {`
`  return x % 2 == 1;`
`}`
 	
`choose a in { 1 .. 8 };`
`if (odd(a)) success;`
	`<k>`
	`  success ;`
	`</k>`
	`<state>`
	`  a |-> 7`
	`</state>`

2.4.3 `failure`

Syntax: `failure ;`

Example:

Algorithm	Output

`odd(x) {`
`  return x % 2 == 1;`
`}`
 	
`s = emptySet;`
`for (i = 0; i < 8; i = i+2)`
`  s.pushBack(i);`
`choose x in s s.t. odd(x);`
	`<k>`
	`  failure ;`
	`</k>`
	`<state>`
	`  i |-> 8`
	`  s |-> { 0, 2, 4, 6 }`
	`  x |-> 6`
	`</state>`

2.5 Functions/Procedures Describing Algorithms

Example:

Algorithm — Output for `n |-> 5 b |-> [5,1,3,2,4]`

```
partition(out a, p, q) {
  x = a[p] ;
  i = p + 1;   j = q;
  while (i <= j) {
    if (a[i] <= x) i = i+1;
    else if (a[j] >= x) j = j-1;
    else if (a[i] > x && x > a[j]) {
      swap(a, i, j);
      i = i+1;
      j = j-1;
    }
  }
  k = i-1;  a[p] = a[k];  a[k] = x;
  return k;
}

qsort(out a, p, q) {
  if (p < q) {
    k = partition(a, p, q);
    qsort(a, p, k-1);
    qsort(a, k+1, q);
  }
}

qsort(b, 0, n-1);
```

```
b |-> [ 1, 2, 3, 4, 5 ]
n |-> 5
```

Note that the output parameters and the input/output parameters are declared with the prefix `out`.